


High-Level Language



Building a Modern Computer From First Principles

www.nand2tetris.org

Where we are at:




Some milestones in the evolution of programming languages

- ❑ Machine language (binary code)
- ❑ Assembly language (low-level symbolic programming)
- ❑ Simple procedural languages, e.g. Fortran, Basic, Pascal, C
- ❑ Simple object-based languages (without inheritance), e.g. early versions of Visual Basic, JavaScript
- ❑ Fancy object-oriented languages (with inheritance): C++, Java, C#



Programming languages

- Procedural programming (e.g. C, Fortran, Pascal)
- Object-oriented programming (e.g. C++, Java, Python)
- Functional programming (e.g. Lisp, ML, Haskell)
- Logic programming (e.g. Prolog)



ML

```
■ fun fac(x) = if x=0 then 1
  else x*fac(x-1);
```

```

■ fun length(L) =
  if (L=nil) then 0
  else 1+length(tl(L));

```

Prolog

- Facts

- `human(kate)`.
- `human(bill)`.
- `likes(bill,kate)`.
- `likes(kate,john)`.
- `likes(john,kate)`.

- Rules

- `friend(X,Y) :- likes(X,Y),likes(Y,X).`

Prolog

- Absolute value

```
abs(X, X) :- X >= 0, !.  
abs(X, Y) :- Y is -X.
```

?- abs(-9,R).

R=9

?- abs(-9,8).


No

- Length of a list


```
my_length([], 0).
```

$$\text{my_length}([_|T], R) :- \text{my_length}(T, R1), R \text{ is } R1+1.$$


?- my_length([a, b, [c, d], e], R).

$$R = 4$$


Procedure oriented programming




Object oriented programming



The Jack programming language

Jack: a simple, object-based, high-level language with a Java-like syntax

Some sample applications written in Jack:



Disclaimer

Although Jack is a real programming language, we don't view it as an *end*.

Rather, we use Jack as a *means* for teaching:

- How to build a compiler
- How the compiler and the language interface with the operating system
- How the topmost piece in the software hierarchy fits into the big picture

Jack can be learned (and un-learned) in one hour.

Roadmap for learning Jack

■ Start with examples

- Hello World
- Procedure and array
- Abstract data types
- Linked list
- ...

■ Formal Jack Spec.

■ More complex examples

Hello world

```
/** Hello World program. */
class Main {
    function void main () {
        // Prints some text using the standard library
        do Output.printString("Hello World");
        do Output.println();      // New line
        return;
    }
}
```

Some observations:

- | | |
|---------------------------|---|
| □ Java-like syntax | □ Class_name.method_name |
| □ Classes | □ Standard library a set of OS services (methods and functions) organized in 8 supplied classes: Math, String, Array, Output, Keyboard, Screen, Memory, Sys |
| □ Entry point: Main.main | |
| □ Typical comments format | |
| □ do for function calls | |

Jack standard library aka language extensions aka Jack OS

```
class Math {
    function void init()
    function int abs(int x)
    function int multiply(int x, int y)
    function int divide(int x, int y)
    function int min(int x, int y)
    function int max(int x, int y)
    function int sqrt(int x)
}
```

Jack standard library aka language extensions aka Jack OS

```
Class String {
    constructor String new(int maxLength)
    method void dispose()
    method int length()
    method char charAt(int j)
    method void setCharAt(int j, char c)
    method String appendChar(char c)
    method void eraseLastChar()
    method int intValue()
    method void setInt(int j)
    function char backSpace()
    function char doubleQuote()
    function char newLine()
}
```

Jack standard library aka language extensions aka Jack OS

```
Class Array {  
    function Array new(int size)  
    method void dispose()  
}
```

```
class Memory {  
    function int peek(int address)  
    function void poke(int address, int value)  
    function Array alloc(int size)  
    function void deAlloc(Array o)  
}
```

Jack standard library aka language extensions aka Jack OS

```
class Output {  
    function void moveCursor(int i, int j)  
    function void printChar(char c)  
    function void printString(String s)  
    function void printInt(int i)  
    function void println()  
    function void backSpace()  
}
```

```
Class Screen {  
    function void clearScreen()  
    function void setColor(boolean b)  
    function void drawPixel(int x, int y)  
    function void drawLine(int x1, int y1, int x2, int y2)  
    function void drawRectangle(int x1, int y1, int x2, int y2)  
    function void drawCircle(int x, int y, int r)  
}
```

Jack standard library aka language extensions aka Jack OS

```
Class Keyboard {  
    function char keyPressed()  
    function char readChar()  
    function String readLine(String message)  
    function int readInt(String message)  
}
```

```
Class Sys {  
    function void halt():  
    function void error(int errorCode)  
    function void wait(int duration)  
}
```

Typical programming tasks in Jack

Jack can be used to develop any app that comes to my mind, for example:

- ❑ Array processing reading/storing numbers in an array
- ❑ Procedural programming: a program that computes $1 + 2 + \dots + n$
- ❑ Object-oriented programming: a class representing bank accounts
- ❑ Abstract data type representation: a class representing fractions (like $2/5$)
- ❑ Data structure representation: a class representing linked lists

We will now discuss the above examples

As we do so, we'll begin to unravel how the magic of a high-level object-based language is delivered by the compiler and by the VM

These insights will serve us in the next lectures, when we build the Jack compiler.

Array example

```
class Main {
  function void main () {
    var Array a;
    var int length;
    var int i, sum;

    let length = Keyboard.readInt("#number:");
    let a = Array.new(length);
    let i = 0;

    while (i < length) {
      let a[i] = Keyboard.readInt("next: ");
      let sum = sum + a[i];
      let i = i+1;
    }

    do Output.printString("The average: ");
    do Output.printInt(sum / length);
    do Output.println();
    return;
  }
}
```

- var: variable declaration
- type: int, Array
- let: assignment
- Array: provided by OS. No type for an array. Actually, it can contain any type and even different types in an array.
- Primitive types: int, boolean, char.
- All types in Jack occupy one word. When declaring a variable of primitive types, the space is reserved. For other types, a reference is reserved.

Procedural programming example

```
class Main {

  /** Sums up 1 + 2 + 3 + ... + n */
  function int sum (int n) {
    var int sum, i;
    let sum = 0;
    let i = 1;
    while (~i > n) {
      let sum = sum + i;
      let i = i + 1;
    }
    return sum;
  }

  function void main () {
    var int n;
    let n = Keyboard.readInt("Enter n: ");
    do Output.printString("The result is: ");
    do Output.printInt(sum(n));
    return;
  }
}
```

Jack program = a collection of one or more classes

Jack class = a collection of one or more subroutines

Execution order: when we execute a Jack program, Main.main() starts running.

Jack subroutine:

- method
- constructor
- function (static method)
- (the example on the left has functions only, as it is "object-less")

Object-oriented programming example

The BankAccount class (skeletal)

```
/** Represents a bank account.
  A bank account has an owner, an id, and a balance.
  The id values start at 0 and increment by 1 each
  time a new account is created. */

class BankAccount {

  /** Constructs a new bank account with a 0 balance. */
  constructor BankAccount new(String owner)

  /** Deposits the given amount in this account. */
  method void deposit(int amount)

  /** Withdraws the given amount from this account. */
  method void withdraw(int amount)

  /** Prints the data of this account. */
  method void printInfo()

  /** Disposes this account. */
  method void dispose()
}
```

Object-oriented programming example (continues)

```
/** Represents a bank account. */
class BankAccount {
  // class-level variable
  static int newAcctId;

  // Private variables(fields/properties)
  field int id;
  field String owner;
  field int balance;

  /** Constructs a new bank account */
  constructor BankAccount new (String owner) {
    let id = newAcctId;
    let newAcctId = newAcctId + 1;
    let this.owner = owner;
    let balance = 0;
    return this;
  }
  // More BankAccount methods.
}
```

```
// Code in any other class:
var int x;
var BankAccount b;
let b = BankAccount.new("joe");
```

Explain `b = BankAccount.new("joe")`
Calls the constructor (which creates a new BankAccount object)

Explain `return this`

The constructor returns the RAM base address of the memory block that stores the data of the newly created BankAccount object

Explain `b = BankAccount.new("joe")`
stores in variable b a pointer to the object's base memory address

Object-oriented programming example (continues)

```

/** Represents a bank account. */
class BankAccount {
  // class-level variable
  static int newAcctId;

  // Private variables(fields/properties)
  field int id;
  field String owner;
  field int balance;

  /** Constructs a new bank account */
  constructor BankAccount new (String
    owner) {
    let id = newAcctId;
    let newAcctId = newAcctId + 1;
    let this.owner = owner;
    let balance = 0;
    return this;
  }
  // More BankAccount methods.
}

```

```

// Code in any other class:
var int x;
var BankAccount b;
let b = BankAccount.new("joe");

```

Behind the scene (following compilation):

```

// b = BankAccount.new("joe")
push "joe"
call BankAccount.new
pop b

```

Explanation: the calling code pushes an argument and calls the constructor; the constructor's code (not shown above; the compiler generates `Memory.alloc(n)` for constructors) creates a new object, pushes its base address onto the stack, and returns;

The calling code then pops the base address into a variable that will now point to the new object.

Object-oriented programming example (continues)

```

class BankAccount {
  static int nAccounts;

  field int id;
  field String owner;
  field int balance;

  // Constructor ... (omitted)

  /** Handles deposits */
  method void deposit (int amount) {
    let balance = balance+amount;
    return;
  }

  /** Handles withdrawals */
  method void withdraw (int amount){
    if (~(amount > balance)) {
      let balance = balance-amount;
    }
    return;
  }
  // More BankAccount methods.
}

```

```

...
var BankAccount b1, b2;
...
let b1 = BankAccount.new("joe");
let b2 = BankAccount.new("jane");
do b1.deposit(5000);
do b1.withdraw(1000);
...

```

Explain `do b1.deposit(5000)`

- In Jack, void methods are invoked using the keyword `do` (a compilation artifact)
- The object-oriented method invocation style `b1.deposit(5000)` is a fancy way to express the procedural semantics `deposit(b1,5000)`

Behind the scene (following compilation):

```

// do b1.deposit(5000)
push b1
push 5000
call BankAccount.deposit

```

Object-oriented programming example (continues)

```

class BankAccount {
  static int nAccounts;

  field int id;
  field String owner;
  field int balance;

  // Constructor ... (omitted)

  /** Prints information about this account. */
  method void printInfo () {
    do Output.printInt(id);
    do Output.printString(owner);
    do Output.printInt(balance);
    return;
  }

  /** Disposes this account. */
  method void dispose () {
    do Memory.deAlloc(this);
    return;
  }
  // More BankAccount methods.
}

```

```

// Code in any other class:
...
var int x;
var BankAccount b;

let b = BankAccount.new("joe");
// Manipulates b...
do b.printInfo();
do b.dispose();

```

Explain

`do Memory.deAlloc(this)`

This is a call to an OS function that knows how to recycle the memory block whose base-address is `this`. We will write this function when we develop the OS (project 12).

Object-oriented programming example (continues)

```

class BankAccount {
  static int nAccounts;

  field int id;
  field String owner;
  field int balance;

  // Constructor ... (omitted)

  /** Prints information about this account. */
  method void printInfo () {
    do Output.printInt(id);
    do Output.printString(owner);
    do Output.printInt(balance);
    return;
  }

  /** Disposes this account. */
  method void dispose () {
    do Memory.deAlloc(this);
    return;
  }
  // More BankAccount methods.
}

```

```

// Code in any other class:
...
var int x;
var BankAccount b;

let b = BankAccount.new("joe");
// Manipulates b...
do b.printInfo();
do b.dispose();

```

Explain

`do b.dispose()`

Jack has no garbage collection; The programmer is responsible for explicitly recycling memory resources of objects that are no longer needed. If you don't do so, you may run out of memory.

Abstract data type example

The Fraction class API (method signatures)

```
/** A fraction consists of a numerator and a denominator, both int values */
class Fraction {
  /** Constructs a fraction from the given data */
  constructor Fraction new(int numerator, int denominator)

  /** Reduces this fraction, e.g. changes 20/100 to 1/5. */
  method void reduce()

  /** Accessors
  method int getNumerator()
  method int getDenominator()

  /** Returns the sum of this fraction and the other one */
  method Fraction plus(Fraction other)

  /** Returns the product of this fraction and the other one */
  method Fraction product(Fraction other)

  /** Prints this fraction */
  method void print()

  /** Disposes this fraction */
  method void dispose()
}
```

Abstract data type example (continues)

```
class Fraction {
  field int numerator, denominator;

  constructor Fraction new (int numerator, int denominator) {
    let this.numerator = numerator;
    let this.denominator = denominator;
    do reduce() // Reduces the new fraction
    return this
  }

  /** Reduces this fraction */
  method void reduce () { // Code omitted }

  // A static method computing the greatest common denominator of a and b.
  function int gcd (int a, int b) { // Code omitted }

  method int getNumerator () {
    return numerator;
  }

  method int getDenominator () {
    return denominator;
  }

  // More Fraction methods follow.
}
```

// Code in any other class:

```
...
var Fraction a, b;
let a = Fraction.new(2,5);
let b = Fraction.new(70,210);
do b.print() // prints "1/3"
...
// (print method in next slide)
```

Abstract data type example (continues)

```
...

// Constructor and previously defined methods omitted

/** Returns the sum of this fraction the other one */
method Fraction plus (Fraction other) {
  var int sum;
  let sum = (numerator * other.getDenominator() +
    (other.getNumerator() * denominator()));
  return Fraction.new(sum , denominator * other.getDenominator());
}


// Similar fraction arithmetic methods follow, code omitted.

/** Prints this fraction */
method void print () {
  do Output.printInt(numerator);
  do Output.printString("/");
  do Output.printInt(denominator);
  return
}

// Code in any other class:
var Fraction a, b, c;
let a = Fraction.new(2,3);
let b = Fraction.new(1,5);
// computes c = a + b
let c = a.plus(b);
do c.print(); // prints "13/15"
```

Data structure example

```
/** Represents a sequence of int values, implemented as a linked list.
The list consists of an atom, which is an int value,
and a tail, which is either a list or a null value. */
```



```
class List {
  field int data;
  field List next;

  /** Creates a new list */
  constructor List new (int car, List cdr) {
    let data = car;
    let next = cdr;
    return this;
  }

  /** Disposes this list by recursively disposing its tail. */
  method void dispose() {
    if (~(next = null)) {
      do next.dispose();
    }
    do Memory.deAlloc(this);
    return;
  }

  ...
} // class List.
```

// Code in any other class:

```
...
// Creates a list holding 2,3, and 5:
var List v;
let v = List.new(5 , null);
let v = List.new(2 , List.new(3,v));
...
```


Jack language specification

- Syntax
- Program structure
- Data types
- Variable kinds
- Expressions
- Statements
- Subroutine calling

(for complete language specification, see the book).

Jack syntactic elements

- A jack program is a sequence of tokens separated by an arbitrary amount of white space and comments.
- Tokens can be symbols, reserved words, constants and identifiers.

```
/** Hello World program. */
class Main {
    function void main () {
        // Prints some text using the standard library
        do Output.printString("Hello World");
        do Output.println();      // New line
        return;
    }
}
```

Jack syntactic elements

White space and comments	Space characters, newline characters, and comments are ignored.	
	The following comment formats are supported:	
	// Comment to end of line	
	/* Comment until closing */	
	/** API documentation comment */	
Symbols	()	Used for grouping arithmetic expressions and for enclosing parameter-lists and argument-lists
	[]	Used for array indexing
	{ }	Used for grouping program units and statements
	,	Variable list separator
	;	Statement terminator
	=	Assignment and comparison operator
	.	Class membership
	+ - * / & ~ < >	Operators
	class, constructor, method, function	Program components
	int, boolean, char, void	Primitive types
Reserved words	var, static, field	Variable declarations
	let, do, if, else, while, return	Statements
	true, false, null	Constant values
	this	Object reference

Jack syntactic elements

Constants	<i>Integer</i> constants must be positive and in standard decimal notation, e.g., 1984. Negative integers like -13 are not constants but rather expressions consisting of a unary minus operator applied to an integer constant.
	<i>String</i> constants are enclosed within two quote (") characters and may contain any characters except <i>newline</i> or <i>double-quote</i> . (These characters are supplied by the functions <code>String.newLine()</code> and <code>String.doubleQuote()</code> from the standard library.)
	<i>Boolean</i> constants can be <code>true</code> or <code>false</code> .
	The constant <code>null</code> signifies a null reference.
Identifiers	Identifiers are composed from arbitrarily long sequences of letters (A-Z, a-z), digits (0-9), and <code>_</code> . The first character must be a letter or <code>_</code> .
	The language is case sensitive. Thus <code>x</code> and <code>X</code> are treated as different identifiers.

Jack program structure

```
class ClassName {  
    field variable declarations;  
    static variable declarations;  
    constructor type { parameterList } {  
        local variable declarations;  
        statements  
    }  
    method type { parameterList } {  
        local variable declarations;  
        statements  
    }  
    function type { parameterList } {  
        local variable declarations;  
        statements  
    }  
}
```

About this spec:

- Every part in this spec can appear 0 or more times
 - The order of the field / static declarations is arbitrary
 - The order of the subroutine declarations is arbitrary
 - Each type is either int, boolean, char, or a class name.
- A Jack program:
- Each class is written in a separate file (compilation unit)
 - Jack program = collection of one or more classes, one of which must be named Main
 - The Main class must contain at least one method, named main()

Jack data types

Primitive types (Part of the language; Realized by the compiler):

- int 16-bit 2's complement (from -32768 to 32767)
- boolean 0 and -1, standing for true and false
- char unicode character ('a', 'x', '+', '%', ...)

Abstract data types (Standard language extensions; Realized by the OS / standard library):

- String
- Array
- ... (extensible)

Application-specific types (User-defined; Realized by user applications):

- BankAccount
- Fraction
- List
- Bat / Ball ... (as needed)

Jack data types

Jack is weakly typed. The language does not define the results of attempted assignment or conversion from one type to another, and different compilers may allow or forbid it.

```
var char c; var String s;  
Let c = 33; // 'A'  
// Equivalently  
Let s = "A"; let c=s.charAt(0);
```

```
var Array a;  
Let a = 5000;  
Let a[100] = 77; // RAM[5100]=77
```

```
var Complex c; var Array a;  
let a = Array.new(2);  
Let a[0] = 7; let a[1] = 8;  
Let c = a; // c==Complex(7, 8)
```

Jack variable kinds and scope

Variable kind	Definition/Description	Declared in	Scope
Static variables	static type name1, name2, ...; Only one copy of each static variable exists, and this copy is shared by all the object instances of the class (like <i>private static variables</i> in Java)	Class declaration.	The class in which they are declared.
Field variables	field type name1, name2, ...; Every object instance of the class has a private copy of the field variables (like <i>private object variables</i> in Java)	Class declaration.	The class in which they are declared, except for functions.
Local variables	var type name1, name2, ...; Local variables are allocated on the stack when the subroutine is called and freed when it returns (like <i>local variables</i> in Java)	Subroutine declaration.	The subroutine in which they are declared.
Parameter variables	type name1, name2, ... Used to specify inputs of subroutines, for example: function void drive (Car c, int miles)	Appear in parameter lists as part of subroutine declarations.	The subroutine in which they are declared.

Jack Statements (five types)

```
let varName = expression;
or
let varName[expression] = expression;
```

```
if (expression) {
    statements
}
else {
    statements
}
```

```
while (expression) {
    statements
}
```

```
do function-or-method-call;
```

```
return expression;
or
return;
```

Jack expressions

A Jack expression is any one of the following:

- A constant
- A variable name in scope (the variable may be static, field, local, or a parameter)
- The keyword `this`, denoting the current object
- An array element using the syntax `arrayName[expression]`, where `arrayName` is a variable name of type `Array` in scope
- A subroutine call that returns a non-void type
- An expression prefixed by one of the unary operators `-` or `~`:
 - `-expression` (arithmetic negation)
 - `~expression` (logical negation)
- An expression of the form `expression op expression` where `op` is one of the following:
 - `+` `-` `*` `/` (integer arithmetic operators)
 - `&` `|` (boolean and and or operators, bit-wise)
 - `<` `>` `=` (comparison operators)
- `( expression )` (an expression within parentheses)

Jack subroutine calls

General syntax: `subroutineName(arg0, arg1, ...)`

where each argument is a valid Jack expression

Parameter passing is *by-value* (primitive types) or *by-reference* (object types)

Example 1:

Consider the function (static method): `function int sqrt(int n)`

This function can be invoked as follows:

```
sqrt(17)
sqrt(x)
sqrt((b * b) - (4 * a * c))
sqrt(a * sqrt(c - 17) + 3)
```

etc. In all these examples the argument value is computed and passed by-value

Example 2:

Consider the method: `method Matrix plus (Matrix other);`

If `u` and `v` were variables of type `Matrix`, this method can be invoked using: `u.plus(v)`

The `v` variable is passed by-reference, since it refers to an object.

Noteworthy features of the Jack language

- The (cumbersome) `let` keyword, as in `let x = 0;`
- The (cumbersome) `do` keyword, as in `do reduce();`
- No operator priority: (language does not define, compiler-dependent)
 - `1 + 2 * 3` yields `9`, since expressions are evaluated left-to-right;
 - To effect the commonly expected result, use `1 + (2 * 3)`
- Only three primitive data types: `int`, `boolean`, `char`;
In fact, each one of them is treated as a 16-bit value
- No casting; a value of any type can be assigned to a variable of any type
- Array declaration: `Array x;` followed by `x = Array.new();`
- Static methods are called function
- Constructor methods are called constructor;
Invoking a constructor is done using the syntax `ClassName.new(argsList)`

Q: Why did we introduce these features into the Jack language?

A: To make the writing of the Jack compiler easy!

Any of these language features can be modified, with a reasonable amount of work, to make them conform to a more typical Java-like syntax.

A simple game: square


- (Demo)
- Use Square as an example.
- Design a class: think of its
 - States: data members
 - Behaviors: function members
- Square
 - `x, y, size`
 - `MoveUp, MoveDown, IncSize, ...`

Perspective

- Jack is an object-based language: no inheritance
- Primitive type system (3 types)
- Standard library
- Our hidden agenda: gearing up to learn how to develop the ...
 - Compiler (projects 10 and 11)
 - OS (project 12).

Principles of object-oriented programming

encapsulation (information hiding)



polymorphism






Fig 6



inheritance

Which language should you learn?




THE LORD OF THE RINGS ANALOGY TO PRO

Python	Java	C	C++	JavaScript
				
Python The Ent	Java Gandalf	C One Ring	C++ Saruman	JavaScript Hobbit
				
Help little Hobbits (beginners) to understand programming concepts Help Wizards (computer scientists) to conduct researches Widely regarded as the best programming language for beginners Easiest to learn Widely used in scientific, technical & academic field, i.e. Artificial Intelligence You can build website using Django, a popular Python web framework	Wants peace & works with everyone (portable) Very popular on all platforms, OS, and devices due to its portability One of the most in-demand & highest paying programming languages Slogaric write once, work everywhere	The power of C is known to them all Everyone wants to get its Power Lingua franca of programming language One of the oldest and most widely used language in the world Popular language for system and hardware programming A subset of C++ except the little details	Everyone thinks that he is the good guy But once you get to know him, you will realize he wants the power, not good deeds Complex version of C with a lot more features Widely used for developing games, industrial and performance-critical applications Learning C++ is like learning how to manufacture, assemble, and drive a car Recommended only if you have a mentor to guide you	Frequently underestimated (powerful) Well-known for the slow, gentle life of the Shire (web browsers) Java and JavaScript are similar like Car and Carpet are similar - Greg Hewgill Most popular clients-side web scripting language A must learn for front-end web developer (HTML and CSS as well) One of the hottest programming language now, due to its increasing popularity as server-side language (node.js)
POPULARITY ★★★★★	POPULARITY ★★★★★	POPULARITY ★★★★★	POPULARITY ★★★★★	POPULARITY ★★★★★
USED TO BUILD YouTube, Instagram, Spotify	USED TO BUILD Gmail, Minecraft, Most Android Apps, Enterprise applications	USED TO BUILD Operating systems and hardware	USED TO BUILD Operating systems, hardware, and browsers	USED TO BUILD Paypal, front-end of majority websites
AVG. SALARY \$107,000	AVG. SALARY \$102,000	AVG. SALARY \$102,000	AVG. SALARY \$104,000	AVG. SALARY \$99,000

ANALOGY TO PROGRAMMING LANGUAGES

JavaScript	C#	Ruby	PHP	Objective-C
				
JavaScript Hobbit	C# Elf	Ruby Man (Middle Earth)	PHP Orc	Objective-C Smaug
				
Frequently underestimated (powerful) Well-known for the slow, gentle life of the Shire (web browsers) Java and JavaScript are similar like Car and Carpet are similar - Greg Hewgill Most popular clients-side web scripting language A must learn for front-end web developer (HTML and CSS as well) One of the hottest programming language now, due to its increasing popularity as server-side language (node.js)	Beautiful creature (language), but stays in their land, Rivendell (Microsoft Platform) A popular choice for enterprise to create websites and Windows application using .NET framework Can be used to build website with ASP.NET, a web framework from Microsoft Similar to Java in basic syntax and some features Learn C# instead of Java if you are targeting to work on Windows platform only	Very emotional creature They (some Ruby developers) feel they are superior & need to rule the Middle Earth Mostly known for its popular web framework, Ruby on Rails Focuses on getting things done Designed for fun and productive coding Best for fun and personal projects, startups, and rapid development	Ugly guy (language) and doesn't respect the rules (inconsistent and unpredictable) Big headache to those (developers) to manage them (coding) Yet still dominates the Middle-earth (most popular web scripting language) Suitable for building small and simple sites within a short time frame Supported by almost every web hosting services with lower price	Lonely and loves gold Primary language used by Apple for Mac OS X & iOS Choose this if you want to focus on developing iOS or OS X apps only Consider to learn Swift (newly introduced by Apple in 2014) as your next language
POPULARITY ★★★★★	POPULARITY ★★★★★	POPULARITY ★★★★★	POPULARITY ★★★★★	POPULARITY ★★★★★
USED TO BUILD Paypal, front-end of majority websites	USED TO BUILD Enterprise and Windows applications	USED TO BUILD Hulu, Groupon, Slideshare	USED TO BUILD WordPress, Wikipedia, Flickr	USED TO BUILD Most iOS Apps and part of Mac OS X
AVG. SALARY \$99,000	AVG. SALARY \$94,000	AVG. SALARY \$107,000	AVG. SALARY \$89,000	AVG. SALARY \$107,000


Programming languages



Most popular PLs (2014/4)

	Apr 2014	Apr 2013	Change	Programming Language	Ratings	Change
1	1			C	17.631%	-0.23%
2	2			Java	17.348%	-0.33%
3	4	3	▲	Objective-C	12.875%	+3.28%
4	3	4	▼	C++	6.137%	-3.58%
5	5			C#	4.820%	-1.33%
6	7	6	▲	(Visual) Basic	3.441%	-1.26%
7	6	7	▼	PHP	2.773%	-2.65%
8	8			Python	1.993%	-2.45%
9	11	10	▲	JavaScript	1.750%	+0.24%
10	12	11	▲	Visual Basic .NET	1.748%	+0.65%
11	10	9	▼	Ruby	1.745%	-0.23%
12	17	16	▲	Transact-SQL	1.170%	+0.45%
13	9	8	▼	Perl	1.027%	-1.31%
14	52	51	▲	F#	0.966%	+0.83%
15	19	18	▲	Assembly	0.853%	+0.14%
16	13	12	▼	Lisp	0.797%	-0.11%
17	18	17	▲	PL/SQL	0.782%	+0.07%
18	24	23	▲	MATLAB	0.760%	+0.24%
19	15	14	▼	Delphi/Object Pascal	0.746%	-0.09%
20	35	34	▲	D	0.708%	+0.39%

Most popular PL trends



Final project

- Assembler for Hack/Toy
- VM translator
- Compiler for Jack
- Finish OS implementation
- Develop applications with Jack
- Design your own computers
- <Fill your ideas here>